

Describing Syntax and Semantics

of

Programming Languages

Part II

Ambiguity

A grammar that generates a sentence for which there are two or more distinct parse trees is said to be **ambiguous**.

Example: ambiguous grammar for simple assignment statements

Consider the string: $A = B + C * A$

$\langle \text{assign} \rangle : \langle \text{id} \rangle = \langle \text{expr} \rangle$

$\langle \text{expr} \rangle : \langle \text{expr} \rangle + \langle \text{expr} \rangle$

$\langle \text{expr} \rangle : \langle \text{expr} \rangle * \langle \text{expr} \rangle$


$\langle \text{expr} \rangle : (\langle \text{expr} \rangle)$

$\langle \text{expr} \rangle : \langle \text{id} \rangle$

$\langle \text{id} \rangle : A$

$\langle \text{id} \rangle : B$

$\langle \text{id} \rangle : C$



ambiguous grammars are problematic

meaning of sentences cannot be determined uniquely ₂

Operator Precedence

Ambiguity in an expression grammar can often be resolved by rewriting the grammar rules to reflect **operator precedence**. This rewrite will involve additional non-terminals and rules.

Modified Grammar

```
<assign> : <id> = <expr>
<expr> : <expr> + <term>
<expr> : <term>
<term> : <term> * <factor>
<term> : <factor>
<factor> : ( <expr> )
<factor> : <id>
<id> : A
<id> : B
<id> : C
```

Leftmost Derivation for $A = B + C * A$

```
<assign>
⇒ <id> = <expr>
⇒ A = <expr>
⇒ A = <expr> + <term>
⇒ A = <term> + <term>
⇒ A = <factor> + <term>
⇒ A = <id> + <term>
⇒ A = B + <term>
⇒ A = B + <term> * <factor>
⇒ A = B + <factor> * <factor>
⇒ A = B + <id> * <factor>
⇒ A = B + C * <factor>
⇒ A = B + C * <id>
⇒ A = B + C * A
```

Unique Parse Tree

Modified Grammar

<assign> : <id> = <expr>

<expr> : <expr> + <term>

<expr> : <term>

<term> : <term> * <factor>

<term> : <factor>

<factor> : (<expr>)

<factor> : <id>

<id> : A

<id> : B


<id> : C

Higher the precedence of operator, the **lower** in the parse tree!

OR

Higher the precedence of operator, the **later** in the grammar rules it appears!

Parse tree for **A = B + C * A**



Operator Precedence continued

The connection between parse trees and derivations is very close; either can easily be constructed from the other.

Every derivation with an unambiguous grammar has a unique parse tree, although that tree can be represented by different derivations.

For example, the derivation of the sentence $A = B + C * A$ (shown to the right) is different from the derivation of the same sentence given previously. But, since the grammar we are using is unambiguous, the parse tree (shown in previous slide) is the same for both derivations.

Rightmost Derivation for $A = B + C * A$

$\langle \text{assign} \rangle$

$\Rightarrow \langle \text{id} \rangle = \langle \text{expr} \rangle$

$\Rightarrow \langle \text{id} \rangle = \langle \text{expr} \rangle + \langle \text{term} \rangle$

$\Rightarrow \langle \text{id} \rangle = \langle \text{expr} \rangle + \langle \text{term} \rangle * \langle \text{factor} \rangle$

$\Rightarrow \langle \text{id} \rangle = \langle \text{expr} \rangle + \langle \text{term} \rangle * \langle \text{id} \rangle$

$\Rightarrow \langle \text{id} \rangle = \langle \text{expr} \rangle + \langle \text{term} \rangle * A$

$\Rightarrow \langle \text{id} \rangle = \langle \text{expr} \rangle + \langle \text{factor} \rangle * A$

$\Rightarrow \langle \text{id} \rangle = \langle \text{expr} \rangle + \langle \text{id} \rangle * A$

$\Rightarrow \langle \text{id} \rangle = \langle \text{expr} \rangle + C * A$

$\Rightarrow \langle \text{id} \rangle = \langle \text{term} \rangle + C * A$

$\Rightarrow \langle \text{id} \rangle = \langle \text{factor} \rangle + C * A$

$\Rightarrow \langle \text{id} \rangle = \langle \text{id} \rangle + C * A$

$\Rightarrow \langle \text{id} \rangle = B + C * A$

$\Rightarrow A = B + C * A$

Associativity

A grammar that describes expressions must handle associativity properly.


The parse tree to the right shows the left addition lower than right addition, indicating left-associativity.

The left-associativity is because of the left-recursion in the first rule for `<expr>`:

`<expr> : <expr> + <term>`

`<expr> : <term>`

To express right-associativity, we can use right-recursive rules.



Right-Associativity (exponent operator)

`<assign> : <id> = <expr>`

`<expr> : <expr> + <term>` ← Left-recursive; left-associative

`<expr> : <term>`

`<term> : <term> * <factor>` ← Left-recursive; left-associative

`<term> : <factor>`

`<factor> : <exp> ** <factor>` ← Right-recursive; right-associative

`<factor> : <exp>`

`<exp> : (<expr>)`

`<exp> : <id>`

`<id> : A`

`<id> : B`

`<id> : C`

$\text{precedence}(**) > \text{precedence}(*) > \text{precedence}(+)$

because + is earlier than * which is earlier than **
in the grammar.



if-else Grammar Rules

`<if_stmt> : IF (<logic_expr>) <stmt>`

`<if_stmt> : IF (<logic_expr>) <stmt> ELSE <stmt>`

`<stmt> : <if_stmt>`

This is an ambiguous grammar!



`if (<logic_expr>) if (<logic_expr>) <stmt> else <stmt>`

An unambiguous Grammar for if-else

The rule for if-else statements in most languages is that an else is matched with the nearest previous unmatched if.

Therefore, between an `if` and its matching `else`, there cannot be an `if` statement without an `else` (an “unmatched” statement).

To make the grammar unambiguous, two new nonterminals are added, representing matched statements and unmatched statements:

```
<stmt> : <matched>
```

```
<stmt> : <unmatched>
```

```
<matched> : if ( <logic_expr> ) <matched> else <matched>
```

```
<matched> : any non-if statement
```

```
<unmatched> : if ( <logic_expr> ) <stmt>
```

```
<unmatched> : if ( <logic_expr> ) <matched> else <unmatched>
```

Attribute Grammars

- An **attribute grammar** can be used to describe more of the structure of a programming language than is possible with a context-free grammar.
- Attribute grammars are useful because some language rules (such as **type compatibility**) are difficult to specify with CFGs.
- Other language rules cannot be specified in CFGs at all, such as the rule that **all variables must be declared before they are referenced**.
- Rules such as these are considered to be part of the **static semantics** of a language, not part of the language's syntax. The term “static” indicates that these rules can be checked at compile time.
- Attribute grammars, designed by **Donald Knuth**, can describe both syntax and static semantics.

Attribute Grammars: continued

- An **attribute grammar** may be informally defined as a context-free grammar that has been **extended** to provide context sensitivity using a set of attributes, assignment of attribute values, evaluation rules, and conditions.
- A finite, possibly empty set of attributes is associated with each distinct symbol in the grammar.
- Each attribute has an associated domain of values, such as integers, character and string values, or more complex structures.
- *Viewing the input sentence (or program) as a parse tree, attribute grammars can pass values from a node to its parent, using a **synthesized attribute**, or from the current node to a child, using an **inherited attribute**.*
- In addition to passing attribute values up or down the parse tree, the attribute values may be assigned, modified, and checked at any node in the derivation tree.
- In particular, attribute grammars add the following to context-free grammars:
 - **Attributes** or properties that can have values assigned to them.
 - **Attribute computation functions** (semantic functions) that specify how attribute values are computed
 - **Predicate functions** that state the semantic rules of the language.

Attribute Grammars: Formal Definition

- An **attribute grammar** is a context-free grammar with the following additional features:
 - A set of attributes $A(X)$ for each grammar symbol X
 - A set of semantic functions and possibly an empty set of predicate functions for each grammar rule
- $A(X)$ consists of two disjoint sets $S(X)$ and $I(X)$, called **synthesized** and **inherited** attributes, respectively

Synthesized attributes are used to pass semantic information up the parse tree

Inherited attributes are used to pass semantic information down and across the tree
- For rule $X_0 : X_1, \dots, X_n$ the **synthesized** attributes for X_0 are computed with a **semantic function** of the form $S(X_0) = f(A(X_1), \dots, A(X_n))$.
- **Inherited** attributes for the symbol X_j , $1 \leq j \leq n$ (in the rule $X_0 : X_1, \dots, X_n$) are computed with a **semantic function** of the form $I(X_j) = f(A(X_0), \dots, A(X_{j-1}))$.

To avoid circularity, inherited attributes are often restricted to functions of the form $I(X_j) = f(A(X_0), \dots, A(X_{j-1}))$.
- A **predicate** function is a Boolean function on the union of attribute sets $A(X_0) \cup \dots \cup A(X_n)$ and a set of literal attribute values. A derivation is allowed to proceed only if all predicates on the rule evaluate to true.

Attribute Grammars: continued

- A parse tree of an attribute grammar is the parse tree based on its underlying CFG, with a possibly empty set of attribute values attached to each node
- If all the attribute values in a parse tree have been computed, the tree is said to be **fully attributed**
- **Intrinsic attributes:** are synthesized attributes of leaf nodes whose values are determined outside the parse tree (coming from the Lexer)
- Initially, the only attributes with values are the intrinsic attributes of the leaf nodes. The semantic functions can then be used to compute the remaining attribute values.

Attribute Grammars: continued

- A parse tree of an attribute grammar is the parse tree based on its underlying CFG, with a possibly empty set of attribute values attached to each node
- If all the attribute values in a parse tree have been computed, the tree is said to be **fully attributed**
- **Intrinsic attributes:** are synthesized attributes of leaf nodes whose values are determined outside the parse tree (coming from the Lexer)
- Initially, the only attributes with values are the intrinsic attributes of the leaf nodes. The semantic functions can then be used to compute the remaining attribute values.

Attribute Grammars: Example 1

The following fragment of an attribute grammar describes the rule that the name on the end of an **Ada**¹ procedure must match the procedure's name:

Syntax Rule: <proc_def> : PROCEDURE PROCNAME[2] <proc_body> END PROCNAME[5] SEMI

Predicate: PROCNAME[2].value == PROCNAME[5].value

Here, we have introduced attribute, called `value`, which is associated with the non-terminal, <proc_name>

Nonterminals that appear more than once in a rule are subscripted to distinguish them.

¹ Ada is the name of a famous programming language from the 70s/80s; used by DOD!

Attribute Grammars: Example 2

Type checking using Attribute Grammars.

Consider the following CFG:

`assign : var = expr`

`expr : var + var`

`expr : var`

`var : A`

`var : B`

`var : C`

with the following conditions:

- Variable types are either **int_type** or **real_type**
- Variables that are added need not be both of the same type. If the types are **different** then the resulting type is **real_type**
- The variable on the LHS of the assignment must have the same type as the expression on the RHS

Attribute Grammars: Example 2 continued

Attributes:

actual_type: A **synthesized** attribute associated with the nonterminals `<var>` and `<expr>`. Stores the actual type (int or real) of a variable or expression. In the case of a variable, the actual type is intrinsic.

expected_type: An **inherited** attribute associated with the nonterminal `<expr>`. Stores the type expected for the expression.

Complete Attribute Grammar

Syntax rule 1: `<assign> : <var> = <expr>`

Semantic rule: `<expr>.expected_type = <var>.actual_type`

Syntax rule 2: `<expr> : <var>[2] + <var>[3]`

Semantic rule:

`<expr>.actual_type = if (<var>[2].actual_type == int_type) and
 (<var>[3].actual_type == int_type)
 then int_type
 else real_type`

Predicate: `<expr>.actual_type == <expr>.expected_type`

Syntax rule 3: `<expr> : <var>`

Semantic rule: `<expr>.actual_type = <var>.actual_type`

Predicate: `<expr>.actual_type == <expr>.expected_type`

Syntax rule 4 `<var> : A`

Semantic rule: `<var>.actual_type = look-up(A.value)`

Syntax rule 5 `<var> : B`

Semantic rule: `<var>.actual_type = look-up(B.value)`

Syntax rule 6 `<var> : C`

Semantic rule: `<var>.actual_type = look-up(C.value)`

The **look-up** function looks up a variable name in the **symbol table** and returns the variable's type.

Attribute Grammars: Example 2 continued

The process of **decorating** the parse tree with attributes could proceed in a completely top-down order if all attributes were inherited.

Alternatively, it could proceed in a completely bottom-up order if all the attributes were synthesized.


Because our grammar has both synthesized and inherited attributes, the evaluation process cannot be in any single direction. One possible order for attribute evaluation is:

- (1) $\langle \text{var} \rangle.\text{actual_type} = \text{look-up}(\text{A})$ (Rule 4)
- (2) $\langle \text{expr} \rangle.\text{expected_type} = \langle \text{var} \rangle.\text{actual_type}$ (Rule 1)
- (3) $\langle \text{var} \rangle[2].\text{actual_type} = \text{look-up}(\text{A})$ (Rule 4)
 $\langle \text{var} \rangle[3].\text{actual_type} = \text{look-up}(\text{B})$ (Rule 4)
- (4) $\langle \text{expr} \rangle.\text{actual_type} = \text{either } \underline{\text{int_type}} \text{ or } \underline{\text{real_type}}$ (Rule 2)
- (5) $\langle \text{expr} \rangle.\text{expected_type} == \langle \text{expr} \rangle.\text{actual_type}$
is either true or false (Rule 2)

Determining attribute evaluation order is a complex problem, requiring the construction of a graph that shows all attribute dependencies.

Consider the assignment


$\text{A} = \text{A} + \text{B}$




Parse Tree

Attribute Grammars: Example 2 continued

The following figure shows the flow of attribute values. Solid lines are used for the parse tree; dashed lines show attribute flow.



The following tree shows the final attribute values on the nodes.



Attribute Grammars: Example 3

Consider the language

$$\{ A^n B^n C^n \mid n > 0 \} = \{ ABC, AABBBCC, AAABBBBCCCC, \dots \}$$

It so happens that there is no CFG for this language!

But, we can devise an attribute grammar for this language.

Step 1: Devise a CFG for the language $\{ A^m B^n C^p \mid m > 0, n > 0, p > 0 \} = \{ ABBBC, ABC, AAABC, \dots \}$

$$\langle s \rangle : \langle a \rangle \langle b \rangle \langle c \rangle$$

$$\langle a \rangle : A \mid \langle a \rangle A$$

$$\langle b \rangle : B \mid \langle b \rangle B$$

$$\langle c \rangle : C \mid \langle c \rangle C$$

Step 2: Extend the CFG by introducing attributes. Introduce synthesized attribute **count** for non-terminals $\langle a \rangle$, $\langle b \rangle$, and $\langle c \rangle$. The semantic functions and the predicate functions are shown in the next slide.

Attribute Grammars: Example 3 continued

Syntax rule 1: $\langle s \rangle : \langle a \rangle \langle b \rangle \langle c \rangle$

Predicate: $\langle a \rangle.\text{count} == \langle b.\text{count} \rangle$ and $\langle b \rangle.\text{count} == \langle c \rangle.\text{count}$

Syntax rule 2: $\langle a \rangle : A$

Semantic rule: $\langle a \rangle.\text{count} = 1$

Syntax rule 3: $\langle a \rangle : \langle a \rangle A$

Semantic rule: $\langle a \rangle[0].\text{count} = \langle a \rangle[1].\text{count} + 1$

Syntax rule 4: $\langle b \rangle : B$

Semantic rule: $\langle b \rangle.\text{count} = 1$

Syntax rule 5: $\langle b \rangle : \langle b \rangle B$

Semantic rule: $\langle b \rangle[0].\text{count} = \langle b \rangle[1].\text{count} + 1$

Syntax rule 6: $\langle c \rangle : C$


Semantic rule: $\langle c \rangle.\text{count} = 1$

Syntax rule 7: $\langle c \rangle : \langle c \rangle C$

Semantic rule: $\langle c \rangle[0].\text{count} = \langle c \rangle[1].\text{count} + 1$

Attribute Grammars: Example 3 continued

Parse tree for AAABBBCCC $\langle a \rangle.\text{count} == \langle b \rangle.\text{count}$ and $\langle b \rangle.\text{count} == \langle c \rangle.\text{count}$



Describing the meaning of programs: Dynamic Semantics

Reasons for creating a formal semantic definition of a language:

- Programmers need to understand the meaning of language constructs in order to use them effectively.
- Compiler writers need to know what language constructs mean to correctly implement them.
- Programs could potentially be proven correct without testing.
- The correctness of compilers could be verified.
- Could be used to automatically generate a compiler.
- Would help language designers discover ambiguities and inconsistencies.

Semantics are typically described in English. Such descriptions are often imprecise and incomplete.

Operational Semantics

Operational semantics describes the meaning of programs by specifying the effects of running it on a machine.

- Using an actual machine language for this purpose is not feasible.
 - The individual steps and the resulting state are too small and too numerous.
 - The storage of a real computer is too large and complex, with several levels of memory (registers, cache, main memory etc)
- **Intermediate-level languages** and interpreters for virtualized computers are used instead.
- Each construct in the intermediate language must have an obvious and unambiguous meaning.
- Operational semantics is the method used in textbooks etc. to describe meaning of programming language constructs!

C for-loop

```
for (expr1; expr2; expr3) {  
    stmts;  
}
```

Meaning

```
expr1;  
loop: if (expr2 == 0) goto out  
    stmts;  
    expr3;  
    goto loop  
out:
```

The human is the virtual computer who is assumed to execute these instructions correctly!

Operational Semantics (continued)

The following statements would be adequate for describing the semantics of the simple control statements of a typical programming language:

```
ident = var
ident = ident + 1
ident = ident - 1
ident = un_op var
ident = var bin_op var
goto label
if (var relop var) goto label
```

where `relop` is a relational operator, `ident` is an identifier, and `var` is either an identifier or a constant.

Adding a few more instructions would allow the semantics of arrays, records, pointers, and subprograms to be described.

Denotational Semantics

- **Denotational semantics**, which is based on recursive function theory, is the most rigorous and most widely known formal method for describing the meaning of programs.
- A denotational description of a language entity is a **function** that maps instances of that entity onto mathematical objects (e.g. numbers, sets of numbers, etc.)
- The term denotational comes from the fact that mathematical objects “denote” the meaning of syntactic entities.
- Each mapping function has a **domain** and a **range**:
 - The syntactic domain specifies which syntactic structures are to be mapped.
 - The range (a set of mathematical objects) is called the semantic domain.

Two Simple Examples

Example 1: Binary numbers

Consider the following grammar to specify string representation of binary numbers:

$\langle \text{bin_num} \rangle : '0'$

$\langle \text{bin_num} \rangle : '1'$

$\langle \text{bin_num} \rangle : \langle \text{bin_num} \rangle '0'$

$\langle \text{bin_num} \rangle : \langle \text{bin_num} \rangle '1'$


The denotational semantic function \mathbf{M}_{bin} maps syntactic objects to nonnegative integers:

$\mathbf{M}_{\text{bin}}('0') = 0$

$\mathbf{M}_{\text{bin}}('1') = 1$

$\mathbf{M}_{\text{bin}}(\langle \text{bin_num} \rangle '0') = 2 * \mathbf{M}_{\text{bin}}(\langle \text{bin_num} \rangle)$


$\mathbf{M}_{\text{bin}}(\langle \text{bin_num} \rangle '1') = 2 * \mathbf{M}_{\text{bin}}(\langle \text{bin_num} \rangle) + 1$



Parse tree for binary string 110

Binary Numbers (continued)

The meanings, or denoted objects (integers in this case), can be attached to the nodes of the parse tree:



$$\begin{aligned} & \mathbf{M_{bin}('110')} \\ &= 2 * \mathbf{M_{bin}('11')} + 0 \\ &= 2 * (2 * \mathbf{M_{bin}('1')} + 1) + 0 \\ &= 2 * (2 * 1 + 1) + 0 \\ &= 2 * 3 + 0 \\ &= 6 \end{aligned}$$

Decimal Numbers

Example 2: Decimal numbers

Grammar rules:

$\langle \text{dec_num} \rangle : '0' \mid '1' \mid '2' \mid '3' \mid '4' \mid '5' \mid '6' \mid '7' \mid '8' \mid '9'$

$\langle \text{dec_num} \rangle : \langle \text{dec_num} \rangle ('0' \mid '1' \mid '2' \mid '3' \mid '4' \mid '5' \mid '6' \mid '7' \mid '8' \mid '9')$

Denotational semantic mappings for these grammar rules:

$$M_{\text{dec}}('0') = 0$$

$$M_{\text{dec}}('1') = 1$$

...

$$M_{\text{dec}}('9') = 9$$

$$M_{\text{dec}}(\langle \text{dec_num} \rangle '0') = 10 * M_{\text{dec}}(\langle \text{dec_num} \rangle)$$

$$M_{\text{dec}}(\langle \text{dec_num} \rangle '1') = 10 * M_{\text{dec}}(\langle \text{dec_num} \rangle) + 1$$

...

$$M_{\text{dec}}(\langle \text{dec_num} \rangle '9') = 10 * M_{\text{dec}}(\langle \text{dec_num} \rangle) + 9$$

$$M_{\text{dec}}('736')$$

$$= 10 * M_{\text{dec}}('73') + 6$$

$$= 10 * (10 * M_{\text{dec}}('7') + 3) + 6$$

$$= 10 * (10 * 7 + 3) + 6$$

$$= 10 * 73 + 6$$

$$= 736$$

State of a Program

- The **state** of a program in denotational semantics consists of the values of the program's variables.
- Formally, the state s of a program can be represented as a set of ordered pairs:

$$s = \{ \langle i_1, v_1 \rangle, \langle i_2, v_2 \rangle, \dots, \langle i_n, v_n \rangle \}$$

where, i_1, \dots, i_n are names of variables and the v_1, \dots, v_n their associated current values.

- Any of the v 's can have the special value **undef**, which indicates that its associated variable is currently undefined.
- Let VARMAP be a function of two parameters, a variable name and the program state. The value of $\text{VARMAP}(i_j, s)$ is v_j .
- Most semantics mapping functions for language constructs map states to states. These state changes are used to define the meanings of the constructs.
- Some constructs, such as expressions, are mapped to values, not states.

Expressions

- In order to develop a concise denotational definition of the semantics of expressions, the following simplifications will be assumed:
 - No side effects.
 - Operators are + and *.
 - At most one operator.
 - Operands are scalar integer variables and integer literals.
 - No parentheses.
 - Value of an expression is an integer.
 - Errors never occur during evaluation; however, the value of a variable may be undefined.
- Here is a CFG for such expressions:

```
<expr> : <dec_num> | <var> | <binary_expr>
<binary_expr> : <left_expr> <operator> <right_expr>
<left_expr> : <dec_num> | <var>
<right_expr> : <dec_num> | <var>
<operator> : + | *
```

Expressions: M_E

```
<expr> : <dec_num> | <var> | <binary_expr>
<binary_expr> : <left_expr> <operator> <right_expr>
<left_expr> : <dec_num> | <var>
<right_expr> : <dec_num> | <var>
<operator> : + | *
```

```
 $M_E$ (<expr>, s) =
  case <expr> of
    <dec_num> =>  $M_{dec}$ (<dec_num>)
    <var> => if (VARMAP(<var>,s) == undef) then error else VARMAP(<var>,s)
    <binary_expr> =>
      if ( $M_E$ (<binary_expr>.<left_expr>,s) == error OR
           $M_E$ (<binary_expr>.<right_expr>,s) == error)
      then error
      elif (<binary_expr>.<operator> == '+')
      then  $M_E$ (<binary_expr>.<left_expr>,s) +  $M_E$ (<binary_expr>.<right_expr>,s)
      else  $M_E$ (<binary_expr>.<left_expr>,s) *  $M_E$ (<binary_expr>.<right_expr>,s)
```


Expressions: M_E Example

Given state $s = \{ (x, 36), (y, 20), (z, 15) \}$

$$\begin{aligned} M_E(x + 24, s) \\ &= M_E(x, s) + M_E(24, s) \\ &= \text{VARMAP}(x, s) + M_{\text{dec}}(24) \\ &= 36 + 10 * M_{\text{dec}}(2) + 4 \\ &= 36 + 10 * 2 + 4 \\ &= 60 \end{aligned}$$

$$\begin{aligned} M_E(x + u, s) \\ &= M_E(x, s) + M_E(u, s) \\ &= \text{VARMAP}(x, s) + M_E(u, s) + 4 \\ &= 36 + M_E(u, s) + 4 \\ &= \text{error} \end{aligned}$$

$$\begin{aligned} M_E(15 + 24, s) \\ &= M_E(15) + M_E(24) \\ &= M_{\text{dec}}(15) + M_{\text{dec}}(24) \\ &= 10 * M_{\text{dec}}(1) + 5 + 10 * M_{\text{dec}}(2) + 4 \\ &= 10 * 1 + 5 + 10 * 2 + 4 \\ &= 39 \end{aligned}$$

Assignment Statement: M_A

```
 $M_A(x = E, s) =$   
  if  $M_E(E, s) == \text{error}$   
  then error  
  else  $s' = \{ \langle i_1', v_1' \rangle, \dots, \langle i_n', v_n' \rangle \},$   
    where for  $j = 1, \dots, n$   
      if  $(i_j == x)$  // note that we are comparing names of variables here  
      then  $v_j' = M_E(E, s)$   
      else  $v_j' = \text{VARMAP}(i_j, s)$ 
```

Note: This mapping for assignment statement does not return any mathematical object, but simply returns a new state s' of the program.

Assignment Statement: M_A Example

Given state $s = \{ (x, 36), (y, 20), (z, 15) \}$

To compute $M_A(x = x + 24, s)$

first calculate meaning of RHS of assignment statement

$$\begin{aligned} & M_E(x + 24, s) \\ &= M_E(x, s) + M_E(24, s) \\ &= \text{VARMAP}(x, s) + M_{\text{dec}}(24) \\ &= \text{VARMAP}(x, s) + 10 * M_{\text{dec}}(2) + 4 \\ &= 36 + 10 * 2 + 4 \\ &= 60 \end{aligned}$$

Then, return the state $s' = \{ (x, 60), (y, 20), (z, 15) \}$

While Loop: M_L

```
 $M_L(\text{while } B \text{ do } L, s) =$   
  if  $M_B(B, s) == \text{error}$  then error  
  elif  $M_B(B, s) == \text{false}$  then s  
  elif  $M_{SL}(L, s) == \text{error}$  then error  
  else  $M_L(\text{while } B \text{ do } L, M_{SL}(L, s))$ 
```

Note:

M_{SL} maps statement lists to states.

M_B maps Boolean expressions to Boolean values (or error).

We assume that these two mappings are defined elsewhere.

While Loop: M_L Example

Consider the statement:

while (**count** > 0) {**x** = **x** + 24; **count** = **count** - 1}
and the initial state **s1** = { (**x**,36), (**count**,1), (**z**,15) }

$M_B((\text{count} > 0), s1) = \text{TRUE}$

$M_L(\text{while}(\text{count} > 0) \{x = x + 24; \text{count} = \text{count} - 1\}, s1)$
 $= M_L(\text{while}(\text{count} > 0) \{x = x + 24; \text{count} = \text{count} - 1\}, s2)$

where, $s2 = M_{SL}(\{x = x + 24; \text{count} = \text{count} - 1\}, s1)$
 $= \{ (x, 60), (\text{count}, 0), (z, 15) \}$

$M_B((\text{count} > 0), s2) = \text{FALSE}$

$M_L(\text{while}(\text{count} > 0) \{x = x + 24; \text{count} = \text{count} - 1\}, s2)$
 $= s2$

Final answer: $s2 = \{ (x, 60), (\text{count}, 0), (z, 15) \}$